<commit_before>{
  description = "githud flake";

  inputs = {
    nixpkgs.url = "github:NixOS/nixpkgs?rev=0bf298df24f721a7f85c580339fb7eeff64b927c";
    flake-utils.url = "github:numtide/flake-utils";
  };

  outputs = { self, nixpkgs, flake-utils }:
    flake-utils.lib.eachDefaultSystem (system:
      let
        pkgs = import nixpkgs {
          inherit system;
        };
      in
      {
        defaultPackage =
          pkgs.haskellPackages.developPackage {
            root = ./.;
            name = "githud";
          };
      });
}<commit_msg>Offer overlay and justStaticExecutables package<commit_after>{
  description = "githud flake";

  inputs = {
    nixpkgs.url = "github:NixOS/nixpkgs?rev=0bf298df24f721a7f85c580339fb7eeff64b927c";
    flake-utils.url = "github:numtide/flake-utils";
  };

  outputs = { self, nixpkgs, flake-utils }:
    let
      overlay = final: prev: {
        haskell = prev.haskell // {
          packageOverrides = hfinal: hprev:
            prev.haskell.packageOverrides hfinal hprev // {
              githud =
                let
                  src = builtins.path {
                    name = "githud-src";
                    path = ./.;
                  };
                in
                hfinal.callCabal2nix "githud" src { };
            };
        };
        githud = final.haskell.lib.justStaticExecutables final.haskellPackages.githud;
      };
    in
    {
      overlays.default = overlay;
    } // flake-utils.lib.eachDefaultSystem (system:
      let
        pkgs = import nixpkgs {
          inherit system;
          overlays = [ self.overlays.default ];
        };
      in
      {
        packages.default = pkgs.githud;
      });
}
{'fraction_non_alphanumeric': 0.11052322163433274, 'fraction_numerical': 0.02880658436213992, 'mean_word_length': 1.659375, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5639506', 'n_tokens_mistral': 610, 'n_tokens_neox': 561, 'n_words': 86}